nCells 5-param double logistic.nb

This *Mathematica* 7 notebook presents, generates data from and retrieves the parameters of a 5-parameter continuous deterministic model of microbial growth and mortality, nCells5[t]. It is constructed from an underlying 3-parameter logistic model of cell's division probability, pDiv5[t], and 3-parameter logistic model of cell's mortality, pMort5[t], having the same steepness parameter, k, but different asymptotes, pdAsym and pmAsym, and characteristic times, tcd and tcm, respectively. As a test, growth/morality data are first generated with this model and then the NonlinearModelFit function is used to fit them and retrieve the values of the 5 original generation parameters. The generated data and the fitted model are then plotted together for visual comparison. Also plotted are the corresponding probability functions pDiv5[t] and pMort5[t]. This is followed by the generation with the same model of some noisy data having a random scatter. The NonlinearModelFit function is used again in an attempt to recover the model's 4 parameters. The generated noisy data and the fitted curve are then plotted together for visual comparison and so are the new probability functions pDiv5[t] and pMort5[t]. The generated noisy data can be replaced by pasted experimental data and the procedure used to estimate the organism's growth/mortality parameters and the corresponding probability functions pDiv5[t] and pMort5[t] under the pertinent conditions.

Programmed by Mark D. Normand based on models developed and elaborated by Micha Peleg, Joseph Horowitz, Murray Eisenberg and Maria G. Corradini. Last modified: November 11, 2009

Clear all variables used in this notebook.

In[593]:=
```
Clear[dt, k, n0, maxNoise, minOffset, nCells5, nCells5Pts, nCells5PtsRand, nlr5Results,
  nParams, nPts5, nPts5Rand, pDiv5, pMort5, pdAsym, pmAsym, plot5FitExact, plot5GenExact,
  plot5FitGenExact, plot5GenRand, plot5FitRand, plot5FitGenRand, plot5pDivpMortExact,
  plot5pDivpMortGen, plot5pDivpMortRand, tcd, tcm, tEn, tSt, yDivMortMax, yMax, yMaxRand]
```

The pDiv5[t] function defines the probability of cell division at time t. The parameters of pDiv5[t] are pdAsym, k and tcd.

In[594]:=
```
pDiv5[t_] := pdAsym / (1. + Exp[k * (tcd - t)])
```

The pMort5[t] function defines the probability of cell mortality at time t. The parameters of pMort5[t] are pmAsym, k and tcm.

In[595]:=
```
pMort5[t_] := pmAsym / (1. + Exp[k * (tcm - t)])
```

n0 is the initial number of cells alive at time t=0.

In[596]:=
```
n0 = 1000.;
```

Define the nCells5Integrate[t] function that combines the pDiv5[t] and pMort5[t] functions to give the number of cells alive at time t.

In[597]:=
```
nCells5Integrate[t_] :=
  n0 * Exp[Subtract @@ (Integrate[pDiv5[tt] - pMort5[tt], tt] /. tt → {t, 0})]
```

If both pDiv5[t] and pMort5[t] can be integrated anlytically, calling nCells5Integrate[t] gives an explicit formula that does NOT include Integrate.

In[598]:= **nCells5Integrate[t]**

Out[598]= $1000.\, e^{\text{pdAsym}\, t - 1.\, \text{pmAsym}\, t - \frac{\text{pdAsym}\, \text{Log}\left[1.+e^{k\, \text{tcd}}\right]}{k} + \frac{\text{pdAsym}\, \text{Log}\left[1.+e^{-1.\, k\, t + k\, \text{tcd}}\right]}{k} + \frac{1.\, \text{pmAsym}\, \text{Log}\left[1.+e^{k\, \text{tcm}}\right]}{k} - \frac{1.\, \text{pmAsym}\, \text{Log}\left[1.+e^{-1.\, k\, t + k\, \text{tcm}}\right]}{k}}$

Now define the nCells5[t] function to be the explicit formula shown above WHICH DOES NOT include Integrate.

In[599]:= **nCells5[t_] = nCells5Integrate[t]**

Out[599]= $1000.\, e^{\text{pdAsym}\, t - 1.\, \text{pmAsym}\, t - \frac{\text{pdAsym}\, \text{Log}\left[1.+e^{k\, \text{tcd}}\right]}{k} + \frac{\text{pdAsym}\, \text{Log}\left[1.+e^{-1.\, k\, t + k\, \text{tcd}}\right]}{k} + \frac{1.\, \text{pmAsym}\, \text{Log}\left[1.+e^{k\, \text{tcm}}\right]}{k} - \frac{1.\, \text{pmAsym}\, \text{Log}\left[1.+e^{-1.\, k\, t + k\, \text{tcm}}\right]}{k}}$

nParams is the number of different parameters in the nCells5[t] model. Assign initial values to all 5 parameter variables. These values will be used to generate a list of {t, nCells5[t]} data values.

In[600]:= **nParams = 5; pdAsym = 0.2; pmAsym = 0.3; k = 0.5; tcd = 10.; tcm = 25.;**

tSt and tEn are the starting and ending times over which to generate and plot the data. dt is the delta-t time increment between data points.

In[601]:= **tSt = 0.; tEn = 40.; dt = 4.;**

minOffset is used in the plots to force plotting the axis tick mark label at 0.

In[602]:= **minOffset = 0.001;**

Assign the maximum y-axis limit of the pDiv5 and pMort5 plots to yDivMortMax.

In[603]:= **pDiv5[tEn]**

Out[603]= 0.2

In[604]:= **pMort5[tEn]**

Out[604]= 0.299834

In[605]:= **yDivMortMax = 0.4;**

Plot the pDiv5[t] and pMort5[t] functions together on the same plot using the currently assigned values of the 5 parameters used to generate the exact data.

In[606]:=
```
plot5pDivpMortGen = Plot[{pDiv5[t], pMort5[t]}, {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True,
  PlotLabel → ToString[nParams] <> "-par. data generation pDiv5[t] (blue) & pMort5[t] (red)",
  FrameLabel → {"time", "pDiv5[t] & pMort5[t]"}]
```

Out[606]=



Generate, fit and plot some exact data (nCells5Pts) using the 5-parameter nCells5[t] function.

Generate some {t, nCells5[t]} data values using the model.

In[607]:=
```
nCells5Pts = Table[{t, nCells5[t]}, {t, tSt, tEn, dt}]
```

Out[607]=
```
{{0., 1000.}, {4., 1016.88}, {8., 1130.32}, {12., 1684.92}, {16., 3353.9}, {20., 7047.47},
 {24., 12 344.8}, {28., 13 151.3}, {32., 9771.52}, {36., 6651.68}, {40., 4468.19}}
```

nPts5 is the number of generated data points in the nCells5Pts list.

In[608]:=
```
nPts5 = Length[nCells5Pts]
```

Out[608]=
```
11
```

Assign the maximum y-axis value to be used in the exact data plots to yMax.

In[609]:=
```
Max[nCells5Pts[[All, 2]]]
```

Out[609]=
```
13 151.3
```

In[610]:=
```
yMax = 14 000.;
```

Clear all the variables to be used by NonlinearModelFit.

In[611]:=
```
Clear[nlr5Results, pdAsym, pmAsym, k, tcd, tcm]
```

Call the NonlinearModelFit function which adjusts the values of the 5 parameters to give the best fit of the nCells5[t] function to the nCells5Pts data and return a report in nlr5Results. All 5 parameters are unconstrained.

In[612]:=
```
nlr5Results = NonlinearModelFit[nCells5Pts, nCells5[t],
  {{pdAsym, 0.20}, {pmAsym, 0.16}, {k, 0.4}, {tcd, 10}, {tcm, 15}}, t, MaxIterations → 500]
```

NonlinearModelFit::sszero :
The step size in the search has become less than the tolerance prescribed by the PrecisionGoal option, but the gradient is larger than the tolerance specified by the AccuracyGoal option. There is a possibility that the method has stalled at a point that is not a local minimum. ≫

Out[612]=
$$\text{FittedModel}\left[\frac{1000.\, e^{\ll 1\gg}\,(1.+\ll 1\gg)^{\ll 19\gg}}{\left(1.+e^{\ll 1\gg}\right)^{\ll 19\gg}}\right]$$

Show the best fitted model, the best fit parameters, the parameter table, the R-squared coefficient of determination, the value of R-squared adjusted for the number of model parameters, the Akaike Information Criterion and the Baysian Information Criterion from the nlr5Results report.

In[613]:=
```
nlr5Results[{"BestFit", "BestFitParameters",
  "ParameterTable", "RSquared", "AdjustedRSquared", "AIC", "BIC"}]
```

Out[613]=
$$\left\{\frac{1000.\, e^{5.49732-0.1\,t}\left(1.+e^{5.-0.5\,t}\right)^{0.4}}{\left(1.+e^{12.5-0.5\,t}\right)^{0.6}},\ \{\text{pdAsym}\to 0.2,\ \text{pmAsym}\to 0.3,\ k\to 0.5,\ \text{tcd}\to 10.,\ \text{tcm}\to 25.\},\right.$$

	Estimate	Standard Error	t Statistic	P-Value
pdAsym	0.2	1.25111×10^{-15}	1.59859×10^{14}	4.04471×10^{-84}
pmAsym	0.3	1.47963×10^{-15}	2.02753×10^{14}	9.71622×10^{-85}
k	0.5	5.15851×10^{-15}	9.69272×10^{13}	8.14012×10^{-83}
tcd	10.	5.02103×10^{-14}	1.99162×10^{14}	1.08159×10^{-84}
tcm	25.	2.21451×10^{-14}	1.12892×10^{15}	3.26088×10^{-89}

$$\left., 1., 1., -522.027, -519.64\right\}$$

Extract the fitted values of the 5 parameters from the nlr5Results report and assign them to the 6 parameter variables.

In[614]:=
```
nlr5Results["BestFitParameters"]
```

Out[614]=
```
{pdAsym → 0.2, pmAsym → 0.3, k → 0.5, tcd → 10., tcm → 25.}
```

In[615]:=
```
pdAsym = nlr5Results["BestFitParameters"][[1, 2]]
```

Out[615]=
```
0.2
```

In[616]:=
```
pmAsym = nlr5Results["BestFitParameters"][[2, 2]]
```

Out[616]=
```
0.3
```

In[617]:=
```
k = nlr5Results["BestFitParameters"][[3, 2]]
```

Out[617]=
```
0.5
```

In[618]:=
```
tcd = nlr5Results["BestFitParameters"][[4, 2]]
```

Out[618]=
```
10.
```

In[619]:=
```
tcm = nlr5Results["BestFitParameters"][[5, 2]]
```

Out[619]=
```
25.
```

plot5GenExact is a ListPlot of the nCells5Pts exact data generated using the 5-parameter nCells5[t] function.

In[620]:=
```
plot5GenExact =
 ListPlot[nCells5Pts, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMax}},
  Frame → True, Axes -> False, PlotStyle → {PointSize → Medium, Blue},
  PlotLabel → ToString[nPts5] <> " nCells5Pts data at dt=" <> ToString[dt] <> " from " <>
    ToString[nParams] <> "-par. model", FrameLabel → {"time", "nCells5Pts data", "", ""}]
```

11 nCells5Pts data at dt=4. from 5−par. model

Out[620]=



plot5FitExact is a Plot of the nCells5[t] function using the currently assigned values of the 5 parameters fitted to the nCells5Pts exact data.

In[621]:=
```
plot5FitExact =
 Plot[nCells5[t], {t, tSt, tEn}, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMax}},
  Frame → True, Axes -> False, PlotStyle → {Blue},
  PlotLabel → ToString[nParams] <> "-par. nCells5[t] curve fitted to nCells5Pts data",
  FrameLabel → {"time", "nCells5[t]", "", ""}]
```

5−par. nCells5[t] curve fitted to nCells5Pts data

Out[621]=



Show the 5-parameter nCells5[t] function curve and the nCells5Pts exact data values together on the same plot.

In[622]:=
```
plot5FitGenExact = Show[plot5FitExact, plot5GenExact,
  PlotLabel → ToString[nPts5] <> " nCells5Pts data at dt=" <>
    ToString[dt] <> " fitted with " <> ToString[nParams] <> "-par. nCells5[t]",
  FrameLabel → {"time", "nCells5Pts data & nCells5[t]", "", ""}]
```

11 nCells5Pts data at dt=4. fitted with 5−par. nCells5[t]

Out[622]=



Plot the pDiv5[t] and pMort5[t] functions together on the same plot using the currently assigned values of the 5 parameters fitted to the nCells5Pts exact data.

In[623]:=
```
plot5pDivpMortExact = Plot[{pDiv5[t], pMort5[t]}, {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True,
  PlotLabel → ToString[nParams] <> "-par. nCells5Pts fit pDiv5[t] (blue) & pMort5[t] (red)",
  FrameLabel → {"time", "pDiv5[t] & pMort5[t]"}]
```

5−par. nCells5Pts fit pDiv5[t] (blue) & pMort5[t] (red)

Out[623]=



Generate, fit and plot some noisy data (nCells5PtsRand) using the 5-parameter nCells5[t] function..

WARNING: Executing the following cell makes *Mathematica* always generate the same repeatable sequence of pseudo-random values from the specified seed value. Run without executing the next cell to restore pseudo-randomness or change the assigned seed value to get a different repeatable pseudo-random sequence.

In[624]:=
```
seed = 0; SeedRandom[seed]
```

maxNoise is the maximum allowed noise value. It must be less than or equal to the initial number of cells, n0.

In[625]:=
```
maxNoise = 500.;
```

nCells5PtsRand is the nCells5Pts list of ordered data pairs in which random noise has been added to the nCells5[t] coordinate values.

If you wish to fit and plot experimental data, assign them to nCells5PtsRand in the same form as shown below.

In[626]:=
```
nCells5PtsRand = Table[{t, nCells5[t] + RandomReal[{-maxNoise, maxNoise}]}, {t, tSt, tEn, dt}]
```

Out[626]=
```
{{0., 1152.47}, {4., 1149.95}, {8., 1313.13}, {12., 1751.27}, {16., 3789.1}, {20., 7523.66},
 {24., 12 083.2}, {28., 13 288.8}, {32., 9372.62}, {36., 6797.21}, {40., 4127.72}}
```

nPts5Rand is the number of generated data points in the nCells5PtsRand list.

In[627]:=
```
nPts5Rand = Length[nCells5PtsRand]
```

Out[627]=
```
11
```

Assign the maximum y-axis value to be used in the noisy data plots to yMaxRand.

In[628]:=
```
Max[nCells5PtsRand[[All, 2]]]
```

Out[628]=
```
13 288.8
```

In[629]:=
```
yMaxRand = 14 000.;
```

Clear all the variables to be used by NonlinearModelFit.

In[630]:=
```
Clear[nlr5ResultsRand, pdAsym, pmAsym, k, tcd, tcm]
```

Call the NonlinearModelFit function which adjusts the values of the 5 parameters to give the best fit of the nCells5[t] function to the nCells5PtsRand noisy data and return a report in nlr5ResultsRand. All 5 parameters are unconstrained.

Note that since the data are dependent on random numbers, the nonlinear fit may be very sensitive to the initial guesses of the values of the 5 parameters.

In[631]:=
```
nlr5ResultsRand = NonlinearModelFit[nCells5PtsRand, nCells5[t],
  {{pdAsym, 0.5}, {pmAsym, 0.5}, {k, 1.}, {tcd, 15.}, {tcm, 20.}}, t, MaxIterations → 500]
```

Out[631]=
```
FittedModel[ 1000. e^«1» (1. + «1»)^«18» / (1. + e^«1»)^«19» ]
```

Show the best fitted model, the best fit parameters, the parameter table, the R-squared coefficient of determination, the value of R-squared adjusted for the number of model parameters, the Akaike Information Criterion and the Baysian Information Criterion from the nlr4Results report.

In[632]:=
```
nlr5ResultsRand[{"BestFit", "BestFitParameters",
  "ParameterTable", "RSquared", "AdjustedRSquared", "AIC", "BIC"}]
```

Out[632]=
$$\left\{\frac{1000.\, e^{5.68053-0.105795\, t}\left(1.+e^{4.10678-0.482159\, t}\right)^{0.369036}}{\left(1.+e^{12.239-0.482159\, t}\right)^{0.588454}},\right.$$

{pdAsym → 0.177934, pmAsym → 0.283728, k → 0.482159, tcd → 8.5175, tcm → 25.3838},

	Estimate	Standard Error	t Statistic	P-Value
pdAsym	0.177934	0.0251777	7.06711	0.0004021
pmAsym	0.283728	0.0321749	8.81831	0.00011808
k	0.482159	0.128977	3.73833	0.00964141
tcd	8.5175	1.28997	6.60284	0.000580342
tcm	25.3838	0.494315	51.3514	3.6593×10^{-9}

$\left., 0.999267, 0.998656, 158.864, 161.252\right\}$

Extract the fitted values of the 5 parameters from the nlr5ResultsRand report and assign them to the 5 parameter variables.

In[633]:=
```
nlr5ResultsRand["BestFitParameters"]
```

Out[633]= {pdAsym → 0.177934, pmAsym → 0.283728, k → 0.482159, tcd → 8.5175, tcm → 25.3838}

In[634]:=
```
pdAsym = nlr5ResultsRand["BestFitParameters"][[1, 2]]
```

Out[634]= 0.177934

In[635]:=
```
pmAsym = nlr5ResultsRand["BestFitParameters"][[2, 2]]
```

Out[635]= 0.283728

In[636]:=
```
k = nlr5ResultsRand["BestFitParameters"][[3, 2]]
```

Out[636]= 0.482159

In[637]:=
```
tcd = nlr5ResultsRand["BestFitParameters"][[4, 2]]
```

Out[637]= 8.5175

In[638]:=
```
tcm = nlr5ResultsRand["BestFitParameters"][[5, 2]]
```

Out[638]= 25.3838

plot5GenRand is a ListPlot of the nCells5PtsRand noisy data generated using the 5-parameter nCells5[t] function.

In[639]:=
```
plot5GenRand =
 ListPlot[nCells5PtsRand, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMaxRand}},
  Frame → True, Axes -> False, PlotStyle → {PointSize → Medium, Red},
  PlotLabel → ToString[nPts5Rand] <> " nCells5PtsRand data at dt=" <>
    ToString[dt] <> " from " <> ToString[nParams] <> "-par. model",
  FrameLabel → {"time", "nCells5PtsRand data", "", ""}]
```

11 nCells5PtsRand data at dt=4. from 5−par. model

Out[639]=



plot5FitRand is a Plot of the nCells5[t] function using the currently assigned values of the 5 parameters fitted to the nCells5PtsRand noisy data..

In[640]:=
```
plot5FitRand = Plot[nCells5[t], {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMaxRand}},
  Frame → True, Axes -> False, PlotStyle → {Red},
  PlotLabel → ToString[nParams] <> "-par. nCells5[t] curve fitted to nCells5PtsRand data",
  FrameLabel → {"time", "nCells5[t]", "", ""}]
```

5−par. nCells5[t] curve fitted to nCells5PtsRand data

Out[640]=



Show the fitted 5-parameter nCells5[t] function curve and the nCells5PtsRand noisy data values together on the same plot.

In[641]:=
```
plot5FitGenRand = Show[plot5FitRand, plot5GenRand,
  PlotLabel → ToString[nPts5Rand] <> " nCells5PtsRand data at dt=" <>
    ToString[dt] <> " fitted with " <> ToString[nParams] <> "-par. nCells5[t]",
  FrameLabel → {"time", "nCells5PtsRand data & nCells5[t]", "", ""}]
```

11 nCells5PtsRand data at dt=4. fitted with 5−par. nCells5[t]

Out[641]=



Plot the pDiv5[t] and pMort5[t] functions together on the same plot using the currently assigned values of the 5 parameters fitted to the nCells5PtsRand noisy data.

In[642]:=
```
plot5pDivpMortRand = Plot[{pDiv5[t], pMort5[t]}, {t, tSt, tEn}, PlotRange →
   {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True, PlotLabel →
   ToString[nParams] <> "-par. nCells5PtsRand fit pDiv5[t] (blue) & pMort5[t] (red)",
  FrameLabel → {"time", "pDiv5[t] & pMort5[t]"}]
```

5−par. nCells5PtsRand fit pDiv5[t] (blue) & pMort5[t] (red)

Out[642]=